

# Partícula en un pozo de potencial simétrico unidimensional

Pablo Santamaría

v0.1 (Mayo 2012)

Consideremos una partícula de masa  $m$  en un pozo de potencial simétrico unidimensional de ancho  $2a$  y profundidad  $V_0 > 0$ :

$$V(x) = \begin{cases} -V_0, & |x| < a \\ 0, & |x| > a \end{cases} \quad (1)$$

Para tal potencial, ilustrado en la fig. 1, la *ecuación de Schrödinger* muestra que la energía total  $E$  en el rango  $-V_0 < E < 0$  sólo puede tomar un conjunto de valores *discretos*, los cuales son las raíces de un par de ecuaciones trascendentes. Una de tales ecuaciones, correspondiente a los denominados *estados pares*, es:

$$\sqrt{\frac{-E}{E + V_0}} = \tan \left( a \sqrt{\frac{2m(E + V_0)}{\hbar^2}} \right), \quad (2)$$

en tanto que la segunda ecuación, correspondiente a los *estados impares*, es:

$$\sqrt{\frac{-E}{E + V_0}} = -\cot \left( a \sqrt{\frac{2m(E + V_0)}{\hbar^2}} \right), \quad (3)$$

donde  $\hbar = \frac{h}{2\pi}$ .

En lo que sigue, sólo consideraremos la resolución de la ecuación (2). Definiendo las constantes positivas, de dimensiones inversa de la longitud,

$$\rho = \sqrt{\frac{-2mE}{\hbar^2}} = \sqrt{\frac{2m|E|}{\hbar^2}} \quad (4)$$

$$k = \sqrt{\frac{2m(E + V_0)}{\hbar^2}} = \sqrt{\frac{2m(V_0 - |E|)}{\hbar^2}} \quad (5)$$

la ecuación para la energía de los estados pares toma la forma

$$\tan(ka) = \frac{\rho}{k}. \quad (6)$$

Puesto que  $\rho$  y  $k$  dependen de  $E$ , esta ecuación sólo puede ser satisfecha para ciertos valores de  $E$ , lo cual establece la *cuantización* de la energía. Para resolverla introduzcamos las cantidades adimensionales

$$\xi = ka, \quad \eta = \rho a. \quad (7)$$





Figura 1. Pozo de potencial simétrico de ancho  $2a$  y profundidad  $V_0$ .



**Figura 2.** Solución gráfica de la ecuación (12). En el caso ilustrado, existen tres soluciones.

Entonces, la ecuación (6) toma la forma:

$$\tan \xi = \frac{\eta}{\xi} \quad (8)$$

Nótese que  $\xi$  y  $\eta$  satisfacen la relación

$$\xi^2 + \eta^2 = \gamma^2, \quad (9)$$

donde

$$\gamma = \sqrt{\frac{2mV_0}{\hbar^2}} a \equiv k_0 a. \quad (10)$$

Se sigue entonces de la relación

$$\frac{1}{\cos^2 \xi} = 1 + \tan^2 \xi = 1 + \left(\frac{\eta}{\xi}\right)^2 = \left(\frac{\gamma}{\xi}\right)^2, \quad (11)$$

que la ecuación (8) es equivalente al sistema de ecuaciones:

$$\begin{cases} |\cos \xi| = \frac{\xi}{\gamma}, \\ \tan \xi > 0. \end{cases} \quad (12)$$

Gráficamente, sus soluciones se encuentran entre los puntos de intersección de la línea recta, de pendiente  $1/\gamma$ , y los arcos cosenoidales, tal como se ilustra en la figura 2. Nótese que si  $\gamma \leq \pi$  la ecuación tiene una única raíz y que cuanto mayor es el valor de  $\gamma$  más soluciones tendrá la misma.

Así pues, para los parámetros del potencial dado, calculado el valor de  $\gamma$  según (10) y se se determinada  $\xi$  como una solución numérica de (12), a continuación se obtiene el correspondiente valor de  $k = \xi/a$ , y, finalmente, el valor de la energía buscado:

$$E = -V_0 + \frac{\hbar^2}{2m} k^2. \quad (13)$$

El siguiente programa Fortran determina los niveles de energía de los estados pares de un electrón ( $m = 9.109 \times 10^{-28}$  gr) en pozo de potencial simétrico cuyos parámetros  $a$  y  $V_0$  son dados en la línea de comandos. El programa hace uso, para resolver numericamente la ecuación planteada, de la subrutina `dfzero` de la biblioteca de rutinas SLATEC a través de la interfaz genérica `zero` implementada en el módulo `slatec95`<sup>1</sup>. Nótese el uso de un módulo para compartir la variable `gamma` que contiene el valor  $\gamma$  calculado en el programa principal con la `function` que define la ecuación del problema. El resultado del programa es una lista de dos columnas, cada fila corresponde a un nivel de energía siendo la primer columna el valor de  $E$  (expresado en erg) y, la segunda columna, su diferencia de energía respecto a  $-V_0$  (expresado en eV).

<sup>1</sup>Ver el apunte *Una interfaz Fortran 95 genérica para las subrutinas fzero/dfzero de SLATEC*.

## Código 1. Cálculo de la energía de los estados pares

```
MODULE common
! -----
! Módulo para compartir el valor de la constante gamma definida por
! los parámetros del pozo de potencial y la masa de la partícula
! -----
USE precision, WP => DP
IMPLICIT NONE
REAL(WP) :: gamma = 0.0_WP
END MODULE common

PROGRAM estados_pares
! -----
! Calculo de la energía de los estados pares de un electrón en un pozo de
! potencial simétrico de profundidad V0 y ancho 2a.
! -----
! Declaración de tipo e inicialización
! -----
USE precision, WP => DP
USE common, ONLY : gamma
USE slatec95
IMPLICIT NONE
! Interface explícita de la función que define la ecuación a resolver
INTERFACE
FUNCTION f(x)
IMPORT :: WP
REAL(WP) :: f
REAL(WP), INTENT(IN) :: x
END FUNCTION f
END INTERFACE
! Parámetros del problema y constantes asociadas
REAL(WP), PARAMETER :: M = 9.109e-28_WP
REAL(WP), PARAMETER :: HBAR = 1.05457172e-27_WP
REAL(WP), PARAMETER :: PI = 3.14159265358979_WP
REAL(WP), PARAMETER :: EV2ERG = 1.60e-12_WP
!
REAL(WP) :: a ! Semiancho del pozo
REAL(WP) :: v0 ! Altura del pozo
REAL(WP) :: e ! Energía
REAL(WP) :: de ! Diferencia de energía respecto de -v0
! Tolerancia para los errores en el cálculo de numérico de las raíces
REAL(WP), PARAMETER :: TOL_REL = 5e-5_WP
REAL(WP), PARAMETER :: TOL_ABS = 0.0_WP
! Otras variables
REAL(WP) :: xl, xr
INTEGER :: i, clave
CHARACTER(32) :: arg
! -----
! Lectura de datos en la línea de comandos: semiancho y profundidad del pozo
! -----
IF (COMMAND_ARGUMENT_COUNT() /= 2) THEN
WRITE(*,*) 'Uso: estados_pares semiancho_a potencial_v0'
STOP
ENDIF
CALL GET_COMMAND_ARGUMENT(1,arg)
READ(arg,*) a
CALL GET_COMMAND_ARGUMENT(2,arg)
READ(arg,*) v0
! -----
! Asignar el valor de gamma e imprimir parámetros del problema
! -----
gamma = sqrt(2.0_WP*M*v0)*A/HBAR
WRITE(*,*) '# a = ', a
WRITE(*,*) '# V0 = ', v0
```

```

WRITE(*,*) '# m = ', m
WRITE(*,*) '# gamma = ', gamma
WRITE(*,*) '# E (ergios)          Delta_E (eV)'
! -----
! Para cada estado calcular la energía encontrando la raíz de la ecuación
! trascendente localizada dentro del intervalo (i*PI,PI/2+i*PI)
! -----
i = 0
DO
  xl = i*PI
  xr = i*PI + PI/2.0_WP
  IF ( xl > gamma ) EXIT
  CALL zero(f,xl,xr,xl,TOL_REL,TOL_ABS,clave)
  IF (clave == 1) THEN
    de = (HBAR*xl/a)**2/(2.0_WP*M)
    e = -v0+de
    WRITE(*,*) e, de/EV2ERG
  ENDIF
  i = i+1
ENDDO
! -----
END PROGRAM estados_pares

FUNCTION f(x)
! -----
! Función que define la ecuación para determinar la energía
! de los estados pares
! -----
USE precision, WP => DP
USE common, ONLY : gamma
IMPLICIT NONE
REAL(WP) :: f
REAL(WP), INTENT(IN) :: x
! -----
f = ABS(COS(x)) - x/gamma
! -----
RETURN
END FUNCTION f

```

La compilación del programa procede como sigue:

```

$ gfortran -Wall -o estados_pares precision.f90 slatec95.f90 \
    estados_pares.f90 -lslatec

```

Consideremos, en particular, la determinación de los niveles de energías de los estados pares de un electrón en el pozo de potencial simétrico dado por:

$$a = 1.95 \times 10^{-8} \text{ cm}$$


$$V_0 = 1.024 \times 10^{-10} \text{ erg} = 64 \text{ eV.}$$

La ejecución del programa conduce a los siguientes resultados:


```

$ ./estados_pares 1.95e-8 1.024e-10
# a      = 1.94999999999999992E-008
# V0     = 1.024000000000000003E-010
# m      = 9.10899999999999972E-028
# gamma  = 7.9865426951238643
# E (ergios)      Delta_E (eV)
-9.92749246750197912E-011  1.9531720781126287
-7.45652425517895173E-011  17.396723405131556
-2.75442120930199667E-011  46.784867441862524

```



**Figura 3.** Energías de los estados pares correspondientes al potencial con  $a = 1.95 \times 10^{-8} \text{ cm}$  y  $V_0 = 1.024 \times 10^{-10} \text{ erg} = 64 \text{ eV}$ .



**Figura 4.** Energías de los estados pares correspondientes al potencial con  $a = 1.95 \times 10^{-8} \text{ cm}$  y  $V_0 = 5 \times 10^{-10} \text{ erg} = 312.5 \text{ eV}$ .

Tenemos así, tres estados discretos de energía, los cuales son ilustrados en la fig. 3. Para el mismo semiancho  $a$  pero con una mayor profundidad, a saber,  $V_0 = 5 \times 10^{-10} \text{ erg}$ , obtenemos seis niveles de energías, los cuales son ilustrados en la figura 4.